EXTENSION DEL METODO DE LOS NUDOS PARA SU APLICACION A CIRCUITOS CON FUENTES PURAS

Quaderns d'enginyeria
2(1980)1 p.51-57

por Miguel Salichs Vivancos

RESUMEN

Cuando se aplica el método de los nudos para el análisis de una red, se condideran implícitas dos condiciones: a) todos los generadores estarán en forma de generadores de corriente con eventual admitancia en paralelo, y b) es posible determinar, mediante cualquier método, la matriz de admitancias de rama. Si el circuito tiene generadores de tensión con impedancia serie. la condición (a) se verifica sin mas que dar su equivalente en forma de generador de corriente y admitancia paralelo; no obstante, si los posibles generadores de tensión tienen impedancia serie nula tal paso no es posible y el método falla. Es en este último caso donde la técnica de análisis que se describe resulta útil, proporcionándonos los potenciales de los nudos independientes así como las corrientes que circulan por los generadores de tensión puros de forma directa.

El artículo finaliza con la exposición de un programa confeccionado en lenguaje BASIC apto para el tratamiento de circuitos resistivos y fuentes constantes.

SUMMARY

Wehen the node voltages method is applied for analyzing a network, two conditions are implicitly considered:

(a) All sources will be as current sources with eventual parallel admittance.

(b) It is possible to determine, by means of any other method, the branch admittance matrix.

If the circuit has voltage sources with series impedance, the (a) condition is verified by giving their equivalents as current sources with parallel amittance. However, if the possible voltage sources have null series impedance this step is not possible and the method fails. It is in this last case where the described analysis technique is useful, it supplies the independent nodes voltages as well as in direct way the circulating currents through the ideal voltage sources.

The article finishes with the exposition of a programme in BASIC language, able to treat resistive circuits and constant sources.

1.- EL METODO DE LOS NUDOS

Se consideraran aquí circuitos sin fuentes controladas; en caso de existir posibles acoplamientos magneticos no serán perfectos. Asi-mismo el análisis se dirige a regímenes permanentes de c.a. senoidal ó bien a circuitos puramente resistivos.

Es sabido que cualquier red eléctrica del tipo mencionado puede caracterizarse mediante las siguientes ecuaciones,

$E_r = Z_r I_r - U_r$ (1.1)

$U_r = A'V$ (1.2)

$AI_r = 0$ (1.3)

en donde

$A(n;r)$ = matriz de incidencia del grafo de la red

$Z_r(r;r)$ = matriz cuadrada (simétrica) de impedancias de rama

$E_r(r;1)$ = matriz columna de ff.ee.mm. de rama

$I_r(r;1)$ = matriz columna de corrientes de rama

$U_r(r;1)$ = matriz columna de dd.dd.pp. de rama

$V\ (r;1)$ = matriz columna de potenciales de nudo independientes.

En el supuesto de ser Zr una matriz invertible ($Y_r = Z_r^{-1}$) resulta de (1.1)

$Y_r E_r = I_r - Y_r U_r$

y con (1.3), (1.2) es,

$-AY_r E_r = (AY_r A')\ V$

Con la notación habitual,

$J_N = -AY_r E_r$ (matriz columna de intensividades de nudo)

$Y_N = AY_r A'$ (matriz simétrica de adminitancias nudo)

se tiene,

$J_N = Y_N V$ ó $V = Y_N^{-1} J_N$ (1.4)

La última ecuación constituye el método de análisis de los nudos. Con (1.4) se hallará fácilmente U_r (1.2) asi como I_r (1.1).

Tal como se ha expuesto el método, aparecen inmediatamente sus limitaciones. En efecto, su éxito estriba en poder determinar $Y_r = Z_r^{-1}$ (#); no obstante en algunos casos, puede escribirse directamente Y_r sin el paso previo de determinar Z_r (ramas con generadores puros de corriente). No se entrará aquí en detalles de la compatibilidad del sistema (1.4) estudiados en múltiples textos (referencia [1]). Existe un caso, de frecuente aplicación práctica,

(#) La condición de ser Y_N invertible pierde aquí su importancia, ya que al ser $Y_N = AY_r A'$ su rango es (Y_r invertible)
rango (Y_N) = rango (A) = n = nº filas ó columnas de Y_N. Asi Y_N es asimismo invertible.

en el cual una ó varias ramas de un esquema eléctrico vienen ocupadas por generadores puros de tensión; aquí el metodo falla estrepitosamente al tener tales ramas impedancias nulas y no existir Z_r^{-1} .

En el caso apuntado puede efectuarse el analisis mediante el método de los bucles o bien recurrir a transformaciones del circuito (eléctricas y topológicas) de forma que la nueva Z_r sea invertible (referencia [1]).

2.- EXTENSION DEL METODO DE LOS NUDOS.

El sistema que se describe aquí no modifica el circuito y es de aplicación general.

Sea r el número total de ramas de un circuito y r_1 aquellas que no estan ocupadas por generadores puros de tensión (enumeradas las r_1 primeras); r_2 las ramas ocupadas por generadores puros de tensión.

Escribiendo (1.1), (1.2) y (1.3) con las particiones adecuadas (subíndices 1 y 2 para ramas sin generadores puros y con generadores puros) es,

$$E_{r_1} = Z_{r_1} I_{r_1} - U_{r_1} \qquad (2.1a)$$

$$E_{r_2} = -U_{r_2} \qquad (2.1b)$$

$$U_{r_1} = A_1' V \qquad (2.2a)$$

$$U_{r_2} = A_2' V \qquad (2.2b)$$

$$A_1 I_{r_1} + A_2 I_{r_2} = 0 \qquad (2.3)$$

Combinando (2.1a), (2.2a) y (2.3) resulta ($Y_{r_1} = Z_{r_1}^{-1}$),

$$-A_1 Y_{r_1} E_{r_1} = A_2 I_{r_2} + (A_1 Y_{r_1} A_1') V$$

y con la notación,

$$J_{N_1} = A_1 Y_{r_1} A_1'$$

(matrices de intensividades y admitancias de nudo "ignorando" las ramas con fuentes puras de tensión) es,

$$Y_{N_1} V + A_2 I_{r_2} = J_{N_1} \qquad (2.4)$$

Analogamente de (2.1b y (2.2b) resulta,

$$A_2' V = -E_{r_2} \qquad (2.5)$$

Las ecuaciones (2.4) y (2.5) resuelven el problema planteado proporcionando además las corrientes I_{r_2} ; escribiendo el sistema de forma compacta,

$$\begin{bmatrix} Y_{N_1} & A_2 \\ A_2' & 0 \end{bmatrix} \begin{bmatrix} V \\ I_{r_2} \end{bmatrix} = \begin{bmatrix} J_{N_1} \\ -E_{r_2} \end{bmatrix} \qquad (2.6)$$

y en consecuencia,

$$\begin{bmatrix} V \\ I_{r_2} \end{bmatrix} = \begin{bmatrix} Y_{N_1} & A_2 \\ A_2' & 0 \end{bmatrix}^{-1} \begin{bmatrix} J_{N_1} \\ -E_{r_2} \end{bmatrix} \qquad (2.7)$$

que junto con (2.2a), (2.1b) y (2.1a) proporcionan las tensiones y corrientes en todas las ramas.

3.- VENTAJA DEL METODO PROPUESTO

Se compara aquí el método propuesto con el de los bucles, en cuanto al orden del sistema a resolver ó matriz a invertir. En el presente estudio (apartado anterior) tal orden es,

$f=n+r_2$

Si b y n son los números de bucles y nudos independientes según la Ley de Weil es,

$r = b + n$

luego,

$b = r - n = (n+r_2) + (r_1 - 2n) = f + (r_1-2n)$

Asi pues el método propuesto presenta un orden menor ó igual al de los bucles si,

$r_1 \geqslant 2n$

4.- CASO PARTICULAR DEL METODO

Supongase que Y_{N_1} es invertible entonces (2.4) y (2.5) proporcionan inmediatamente

$$I_{r_2} = (A_2' Y_{N1}^{-1} A_2)^{-1} \left\{ E_{r_2} + A_2' Y_{N_1}^{-1} J_{N_1} \right\} \qquad (4.1)$$

$$V = Y_{N_1}^{-1} \left\{ J_{N_1} - A_2 I_{r_2} \right\} \qquad (4.2)$$

Observese que $A_2' Y_{N_1}^{-1} A_2$ tiene asimismo inversa, ya que, rango$(A_2' Y_{N_1}^{-1} A_2)$ = rango $(A_2) = r_2$ (si rango$(A_2) < r_2$ ello indicaría la presencia de r_2 - rango (A_2) bucles formados exclusivamente con generadores puros de tensión

Resulta interesante, de cara a las aplicaciones, escribir (4.1) y (4.2) en el supuesto de existir solamente generadores puros de tensión ($J_{N_1} = 0$); entonces,

$$I_{r_2} = (A_2' Y_{N_1}^{-1} A_2)^{-1} E_{r_2} \qquad (4.3)$$

$$V = -Y_{N_1}^{-1} A_2 I_{r_2} \qquad (4.4)$$

5.- PROGRAMA Y EJEMPLO

Para circuitos resistivos se da un programa escrito en BASIC que actua con la técnica descrita. Los datos a entrar son primeramente,

n, r_1, r_2

Para cada rama r_1 con posible generador de tensión dar los 4 números siguientes:

(nudo+), (nudo-), resistencia, f.e.m.

Si el generador fuera de corriente,

-(nudo+), -(nudo-), Conductancia, intensividad

Para ramas r_2 especificar 3 números,

nudo+, nudo-, f.e.m.

PROGRAMA

```
10 DIM M[20,20],K[25,20],J[20],L[20],E[30],G[25],S[25]
20 DISP "N,R1,R2";
30 INPUT N,R1,R2
40 R=R1+R2
50 REDIM M[N+R2,N+R2],K[R1,N],J[N+R2],L[N+R2],E[R],G[R1],S[R1]
60 MAT M=ZER
70 MAT J=ZER
80 MAT K=ZER
90 MAT S=ZER
100 PRINT "DATOS (N="N;"R1="R1;"R2="R2")"
110 PRINT
120 FOR I=1 TO R1
130 DISP "DATOS RAMA"I;
140 INPUT U,V,R,E[I]
150 PRINT I;U;V;R,E[I]
160 IF (U+V)<0 THEN 210
170 S[I]=1
180 G[I]=1/R
190 F=E[I]/R
200 GOTO 260
210 S[I]=-1
220 G[I]=R
230 F=E[I]
240 U=ABS(U)
250 V=ABS(V)
260 IF U*V=0 THEN 280
270 M[U,V]=M[V,U]=M[U,V]-G[I]
280 IF U=0 THEN 330
290 M[U,U]=M[U,U]+G[I]
300 J[U]=J[U]-F
310 K[I,U]=1
320 IF V=0 THEN 360
330 M[V,V]=M[V,V]+G[I]
340 J[V]=J[V]+F
350 K[I,V]=-1
360 NEXT I
370 FOR H=1 TO R2
380 I0=N+H
390 I=H+R1
400 DISP "DATOS RAMA"I;
410 INPUT U,V,E[I]
420 PRINT I;U;V;"0",E[I]
430 IF U=0 THEN 460
440 M[I0,U]=M[U,I0]=1
450 IF V=0 THEN 470
460 M[I0,V]=M[V,I0]=-1
470 J[I0]=-E[I]
480 NEXT H
490 PRINT
500 PRINT "************************************************************"
510 PRINT
520 MAT M=INV(M)
530 MAT L=M*J
540 PRINT "POTENCIALES NUDOS INDEPENDIENTES"
550 PRINT
560 FOR I=1 TO N
570 PRINT I,L[I]
580 NEXT I
590 PRINT
600 PRINT "DD.DD.PP. Y CORRIENTES RAMAS"
610 PRINT
620 FOR I=1 TO R1
630 A=0
640 FOR P=1 TO N
650 A=A+K[I,P]*L[P]
660 NEXT P
670 A0=0.5*(1+S[I])*(E[I]+A)*G[I]+0.5*(1-S[I])*(E[I]+A*G[I])
680 PRINT I,A,A0
690 NEXT I
700 FOR H=1 TO R2
710 I0=N+H
720 I=H+R1
730 PRINT I,-E[I];L[I0]
```

```
740 NEXT H
750 PRINT
760 PRINT "*************************************************************************"
770 PRINT "*************************************************************************"
780 PRINT
790 PRINT
800 END
```

EJEMPLO





En la red de la figura se tiene, $r_1=9$; $r_2=1$; $n=4$. El método propuesto es ventajoso ($r_1 > 2n$). Resulta,

(Los nudos se han enumerado asignando el $\boxed{0}$ al de referencia.)

POTENCIALES NUDOS INDEPENDIENTES

1	-100
2	-34.64760966
3	-65.05667817
4	-76.73730902

DD.DD.PP. Y CORRIENTES RAMAS

1	-34.94332183	-3.494332183
2	-65.35239034	-16.33809759
3	30.4090685	6.081813701
4	23.26269098	2.232626910
5	-34.64760966	-3.464760966
6	11.68063085	5.840315427
7	-65.05667817	-3.252833908
8	76.73730902	15.3474618
9	-42.08969936	18.95515032
10	-100	22.06505668

****
****

BIBLIOGRAFIA

1) Shu- Park Chan (1969): Introductory topological analysis of electrical networks. HRW series in electrical engineering, electronics, and systems.

2) Mayer, D (1970): A contribution to the generalized formulation of the matrix methods of mesh currents and node voltages. Aplikace Matematiky, fascículo 4 pags.255/262. Academia cnecoslovaca de ciencias.